

## LABORATORIO EL TOPO GOLPEADO

The screenshot displays the Android Studio development environment. On the left, the 'Preview' window shows a mobile application titled 'EL JUEGO DEL TOPO'. The app's interface includes a status bar at the top with the time 9:48, a title bar, a large white play area with a small top character icon, and a bottom control panel with buttons for 'REINICIAR', '+', '-', and 'FINALIZAR'. Below these are score indicators for 'Aciertos: 0' and 'Fallos: 0', a level indicator 'NIVEL: 5', and a text line 'Nivel 5: Normal Nivel 1: Experto Nivel 8: Principiante'. The bottom of the screen shows the standard Android navigation bar. Below the preview is a 'Non-visible components' section with icons for 'Clock1' and 'Sound1'.

The 'Components' panel on the right shows a hierarchical tree of the app's UI elements:

- Screen1
  - Canvas1
    - Topo
  - HorizontalArranger
    - Reiniciar
    - Incrementar
    - Reducir
    - Finalizar
  - HorizontalArranger
    - Label1
    - Aciertos
    - Label3
    - Fallos
  - HorizontalArranger
    - Label4
    - Nivel
    - Label5

The 'Properties' panel on the right shows the configuration for the selected 'Screen1' component:

- Screen1
  - AboutScreen: [Text Field]
  - AlignHorizontal: Left : 1
  - AlignVertical: Top : 1
  - AppName: uvprueba30
  - BackgroundColor: White
  - BackgroundImage: None...
  - CloseScreenAnimation: Default
  - Icon: None...
  - OpenScreenAnimation: Default
  - ScreenOrientation: Unspecified
  - Scrollable: [Checked]
  - ShowListsAsJson: [Checked]

The 'Media' panel at the bottom right shows a file named 'topo.png' with an 'Upload File ...' button.

```
to moverTopo
do call Topo . MoveTo
  x random integer from 0 to Canvas1 . Width - Topo . Width
  y random integer from 0 to Canvas1 . Height - Topo . Height
```

```
when Screen1 . Initialize
do call moverTopo
```

```
when Clock1 . Timer
do call moverTopo
```

```
when Topo . Touched
x y
do call Sound1 . Vibrate
  millisecs 100
```

```
when Finalizar . Click
do close application
```

```
when Incrementar . Click
do set Clock1 . TimerInterval to Clock1 . TimerInterval - 36
  set Nivel . Text to Nivel . Text - 1
```

```
when Canvas1 . Touched
x y touchedAnySprite
do if get touchedAnySprite
  then set Aciertos . Text to Aciertos . Text + 1
  else set Fallos . Text to Fallos . Text + 1
```

```
when Reducir . Click
do set Clock1 . TimerInterval to Clock1 . TimerInterval + 36
  set Nivel . Text to Nivel . Text + 1
```

```
when Reiniciar . Click
do set Aciertos . Text to 0
  set Fallos . Text to 0
  set Nivel . Text to 5
```